

# D.C. Hayes Associates, Inc.

## MICROCOMPUTER PRODUCTS

### Micromodem II Application Note No.1

#### Getting into Terminal mode successfully from BASIC

From the keyboard:

```
PR#0          (be sure output goes to screen)
IN#3          (if Micromodem II is in slot 3)
CTRL-A CTRL-F (enter terminal mode in full duplex)
```

From a BASIC program: (assume DOS is loaded)

```
5 D$=CHR$(4) : REM D$=CONTROL D
10 PRINT D$;"PR#0"
20 PRINT D$;"IN#3"
30 POKE 2043,9 : REM FSW see page 37
40 POKE 1915,138 : REM ENTER TERM MODE see page 73
```

#### Automatic log-on to a time share computer

Some Micromodem II users have had problems writing programs designed to automatically dial and log-on to a time share computer system. The task seems very simple yet these programs refuse to work. The root of the problem lies in timing. All modems have a carrier detection delay time which can be a few milliseconds to several seconds. When the Micromodem II firmware has finished dialing it waits for the other modem to send a carrier. When it detects the carrier it turns on its own carrier and instantly returns to BASIC where the next line will execute. Here is where we have the timing problem. The other modem has NOT detected our carrier yet. We must wait for 2 to 4 seconds before trying to send any data to the other modem. The following program fragment will show how this is done.

```
10 D$=CHR$(4) : REM CONTROL D FOR DOS
30 PRINT D$;"PR#3"
40 PRINT CHR$(17);"394-4220"
50 FOR X=1 TO 3000
60 NEXT X          : REM 4 SECOND DELAY
70 POKE 2043,9 : REM 7 BITS, EVEN PARITY see page 37
80 PRINT "PASS WORD"
90 PRINT D$;"PR#0"
100 PRINT D$;"IN#3" : REM ENABLE FOR INPUT
110 INPUT A$       : REM GET REPLY
```

The "PASS WORD" and the remainder of the program will depend on the users application. Also, you should never have a "PR#3" and a "IN#3" in effect at the same time in this application.

#### Special note for Apple Serial Interface card owners

The Apple Serial Interface card (part # 600-0029) has an obscure bug in its firmware which causes it to malfunction when any other interface card with more than 256 bytes of PROM is installed in the next higher slot. For example if the Serial interface card is in slot 2 and the Micromodem II is in slot 3 the Serial Interface card will not work. You should install the Serial Interface card on the left side of an empty slot or one containing a card with 256 or less bytes of PROM memory on it.

```

11400 R = R + 1
11500 IF R < 128 THEN GOTO 10500
11600 PRINT "TEST COMPLETED"
11700 PRINT E;" ERRORS DETECTED"
11800 RETURN
20000 POKE MDEM,8
20050 POKE ACIA,3
20100 IF PEEK (ACIA) < 4 THEN GOTO 20100
20150 PRINT : PRINT : PRINT "TESTING ";
20210 MODE$ = "ORIGINATE"
20220 IF M = 1 THEN GOTO 20240
20230 MODE$ = "ANSWER"
20240 BAUD = 300
20250 IF B = 1 THEN GOTO 20270
20260 BAUD = 110
20270 PRINT BAUD;" BAUD ";MODE$;" MODE "
20300 POKE MDEM,154 + 4 * M + B
20400 M = M < > 1
20500 IF M < > 0 THEN GOTO 20700
20600 B = B < > 1
20700 DLY = 0
20710 DLY = DLY + 1
20720 IF PEEK (ACIA) < 4 THEN RETURN
20730 IF DLY < 500 THEN GOTO 20710
20740 PRINT "NO CARRIER DETECT"
20750 E = E + 1: RETURN
20800 RETURN
30000 REM EPROM TEST ROUTINE
30050 PRINT : PRINT
30100 PRINT "STARTING EPROM TEST"
30200 PRINT "PLEASE WAIT 20 SECONDS"
30300 RB = 778: REM BYTE READ FROM EPROM
30400 POKE 768,173: REM LDA
30500 POKE 769,0: REM 00
30600 POKE 770,192 + SLOT
30700 POKE 771,173: REM LDA
30800 POKE 774,141: REM STA
30900 POKE 775,10: REM 0A
31000 POKE 776,3: REM 03
31100 POKE 777,96: REM RTS
32000 A = PEEK ( - 12289): REM TURN OFF OTHER ROMS
32100 X = 0: REM CLEAR CHECKSUM
32200 FOR H = 200 TO 203
32300 FOR L = 0 TO 255
32400 POKE 772,L: POKE 773,H
32500 CALL 768:X = X + PEEK (RB)
32600 NEXT L
32700 NEXT H
32800 CALL - 198
32900 IF X = CK THEN PRINT "EPROM IS OK"
33000 IF X < > CK THEN PRINT "EPROM IS DEFECTIVE"
33100 END
40000 REM OUR THANKS TO ROBERT J. DENISON
40100 REM FOR APPLESOFT CONVERSION.

```

Micromodem II is a trademark of D.C. Hayes Associates, Inc.

line 20700. This allows about 8 seconds for carrier detect before printing "NO CARRIER DETECT". Also, on line 10620, some code has been added to detect the complete lack of received characters. It now prints "NO DATA RECEIVED" instead of hanging forever. Lines 30000 to 33100 are an EPROM test program. This code will test the 1K EPROM or ROM which holds the Micromodem II firmware. It works by adding up all the bytes and comparing the result with the known correct value for the Rev. K EPROM which is 122440.

Each phase of the test takes about 10 seconds except the EPROM test which takes 20.

```
5  REM  SELFTEST PROGRAM IN APPLESOFT
10  CK = 122440: REM  REV K EPROM CHECKSUM
20  HOME : VTAB (4)
30  PRINT "SELF-TEST PROGRAM FOR THE"
40  PRINT "D.C.HAYES ASSOCIATES,INC"
50  PRINT "      MICROMODEM II"
60  PRINT : PRINT
100 INPUT "WHICH SLOT IS THE MICROMODEM II IN ? ";SLOT
110 PRINT : PRINT "UNPLUG THE MICROMODEM II"
112 PRINT "FROM THE MICROCOUPLER": PRINT
115 INPUT "PRESS RETURN WHEN THIS IS DONE ";MODE$
200 DTA = - 16249 + 16 * SLOT
300 ACIA = DTA - 1:MDEM = DTA - 2:CHAR = 1912
400 T = 0:M = 0:B = 0: HOME
600  REM  SET UP MODEM FOR TEST
700  GOSUB 20000
800  REM  PERFORM TEST
900  GOSUB 10000
1000 T = T + E
1100 IF M OR B = 1 THEN GOTO 700
1200 POKE MDEM,8: PRINT : PRINT
1300 PRINT "HARDWARE TEST COMPLETE"
1400 PRINT T;" ERRORS DETECTED"
1500 GOTO 30000: REM  NOW TEST EPROM
10000 PRINT "COMMENCING TEST"
10200 E = 0
10300 POKE ACIA,3
10400 POKE ACIA,21
10450 S = 0:R = 0
10500  REM
10550 L = PEEK (ACIA)
10551 LMOD = L - ( INT (L / 2) * 2)
10552 IF LMOD = 1 THEN GOTO 11100
10600 IF PEEK (ACIA) < 2 THEN GOTO 10500
10620 IF (S - R) < 4 THEN GOTO 10700
10630 PRINT "NO DATA RECEIVED"
10650 E = E + 1: RETURN
10700 IF S > 127 THEN GOTO 10500
10800 POKE CHAR,S
10850 CALL - 16382 + 256 * SLOT
10900 S = S + 1
11000 GOTO 10500
11100 X = PEEK (DTA)
11200 IF X = R THEN GOTO 11400
11300 E = E + 1
```

*PHYSICAL  
SETUP*

*INIT*

## Remote console operation with the disk

We have received many calls from people who wanted to call and use their Apple computer from a remote location. There has been much confusion about which options to use and how to bring up DOS without disabling the Micromodem II. The following program should take care of most of the problems. It should replace your present INIT or HELLO program on the disk so that it will run each time the DOS is booted up. It will take care of all modem initialization and option settings.

Line 20 sets the Micromodem II slot number

Line 25 sets the data word format. See page 37 for details.

Line 110 sets a 70 millisecond delay after the CR-LF sequence to allow the print head on printing terminals to return.

Line 170 sets the following options:

- Line feeds after carriage returns
- Transparent mode (control characters won't affect modem)
- Local display and keyboard are enabled

```
1 REM THIS IS A "HELLO" PROGRAM
2 REM FOR THE APPLE II
3 REM AND THE MICROMODEM II
4 REM MODEM IS IN SLOT 3
10 TEXT
20 SLOT=3: REM MODEM SLOT #
25 FSW=9: REM 7 DATA BITS, EVEN PARITY
30 DS="": REM DS = CNTRL D
40 PRINT DS;"NOMONI,O,C"
50 CALL -936
60 VTAB 10
70 PRINT DS;"IN#";SLOT
80 PRINT DS;"PR#";SLOT
85 REM
90 REM SET FOR 70 MS DELAY
100 REM AFTER CR/LF SEQUENCE
110 POKE 1528+SLOT,7
115 REM
120 REM SET OPTIONS ON MODEM
130 REM LINE FEED INSERT
140 REM TRANSPARENT MODE
150 REM LOCAL DISPLAY
160 REM LOCAL KEYBOARD ENABLE
170 POKE 1912+SLOT,16+4+2
175 POKE 2040+SLOT,FSW
180 PRINT "MICROMODEM II SETUP FOR REMOTE CONSOLE"
200 END
```

*CR DELAY*

*SET FLAGS*

*SET ACIA WORD*

## An improved self-test program for the Micromodem II

This is an Applesoft version of the self-test program on page 10 of the Micromodem II manual. It works as described on pages 9 and 10 but has some additional features. A time-out has been added to the wait for carrier logic on